
To demonstrate the QIBM_RPG_CHECK_FILES* envvars 
 
This demonstration takes advantage of the situation where doing RCLRSC can cause MCH3402 when files are opened by a *CALLER service 
running in the default activation group. 
 
The RCLRSC causes the files to be closed, but it does not reinitialize the storage used by the modules in the service program, so the modules 
think the file is still open. This causes MCH3402 when an I/O operation is attempted in the module after the file was closed due to the 
RCLRSC. 
 

Source for an ILE RPG module MCH3402TST 
       ctl-opt nomain;  
 
       dcl-f qrpglesrc disk(92) extfile('QGPL/QRPGLESRC'); // Global file 
       dcl-ds ds len(92) end-ds;  
 
       dcl-proc p1 export;  
          dcl-f qrpgsrc disk(92) static            // Static local file 
                        extfile('QGPL/QRPGSRC');  
          dcl-s calls int(10) static; 
 
          calls += 1; 
          snd-msg 'Start P1, this is call number ' + %char(calls);   
          snd-msg '... read local file QRPGSRC next'; 
          read qrpglesrc ds;  
          snd-msg '... read global file QRPGLESRC next'; 
          read qrpgsrc ds;  
          snd-msg '... Return from P1'; 
       end-proc;  

 
Compile with CRTRPGMOD and with the compile-time environment variable 

 
==> ADDENVVAR QIBM_RPG_CHECK_FILES VALUE('Y') LEVEL(*JOB) 
==> CRTRPGMOD QTEMP/MCH3402TST SRCFILE(...) 

 

Test 1: Without the runtime environment variable 
DLTSRVPGM QTEMP/MCH3402T1 /* Ignore the error if it doesn't exist */ 
 
CRTSRVPGM QTEMP/MCH3402T1 MODULE(QTEMP/MCH3402TST) EXPORT(*ALL) ACTGRP(*CALLER) 
 
RMVENVVAR QIBM_RPG_CHECK_FILES_ACTION LEVEL(*JOB) /* Ignore the error if it doesn’t exist */ 
 
/* QZRUCLSP is an API that calls a procedure in a service program. X’D7F100’ is “P1” null-terminated */ 
CALL PGM(QZRUCLSP) PARM(('MCH3402T1 QTEMP') (X'D7F100') 
      (0 (*INT 4)) (0 (*INT 4)) (0 (*INT 4)) (x'00000000'))  
/* Notice that it says it was call 1 and that it will read QRPGSRC and QRPGLESRC*/ 
 
DSPJOB OPTION(*OPNF) /* Notice that QRPGLESRC and QRPGSRC are open */ 
 
RCLRSC 
 
DSPJOB OPTION(*OPNF) /* Notice that QRPGLESRC and QRPGSRC are not open */ 
 
CALL PGM(QZRUCLSP) PARM(('MCH3402T1 QTEMP') (X'D7F100') 
      (0 (*INT 4)) (0 (*INT 4)) (0 (*INT 4)) (x'00000000')) 
/* Notice that it says it was call 2 and that it will read QRPGSRC next */ 
 
The joblog will show the following MCH3402 message after the RPG program puts out its messages about starting 
call 2 and reading QRPGSRC next 

Tried to refer to all or part of an object that no longer exists. 

Test 2: With the runtime environment variable, with option EXCEPTION 
DLTSRVPGM QTEMP/MCH3402T2 /* Ignore the error if it doesn't exist */ 
 
CRTSRVPGM QTEMP/MCH3402T2 MODULE(QTEMP/MCH3402TST) EXPORT(*ALL) ACTGRP(*CALLER) 
 
ADDENVVAR QIBM_RPG_CHECK_FILES_ACTION VALUE('EXCEPTION;') REPLACE(*YES) LEVEL(*JOB) 
 
CALL PGM(QZRUCLSP) PARM(('MCH3402T2 QTEMP') (X'D7F100') 
      (0 (*INT 4)) (0 (*INT 4)) (0 (*INT 4)) (x'00000000'))  
 
DSPJOB OPTION(*OPNF) /* Notice that QRPGLESRC and QRPGSRC are open */ 
 
RCLRSC 
 
DSPJOB OPTION(*OPNF) /* Notice that QRPGLESRC and QRPGSRC are not open */ 
 
CALL PGM(QZRUCLSP) PARM(('MCH3402T2 QTEMP') (X'D7F100') 
      (0 (*INT 4)) (0 (*INT 4)) (0 (*INT 4)) (x'00000000')) 
/* It will get a file-check error when it tries to do the first READ */ 
 
The joblog will show the following after the RPG program puts out its message about reading QRPGSRC next. The 
first message is a diagnostic message and the last message is an escape message. 

QIBM_RPG_CHECK_FILES_ACTION Error detected.                               
Error detected in procedure P1 in QTEMP/MCH3402T2/MCH3402TST at statement 
  14 for event 'before_io'                                                
Global file QRPGLESRC: File is closed but RPG open-flag is ON             
Local file QRPGSRC: File is closed but RPG open-flag is ON                
   The previous 4 checks were correct, at the following statements        
      Statement 14, event before_io                                       
      Statement 14, event after_io                                        
      Statement 16, event before_io                                       
      Statement 16, event after_io                                        
QIBM_RPG_CHECK_FILES_ACTION Error detected. See previous messages for     
  details.                                                                

 


